

VLSI Design

Tsung-Chu Huang

Department of Electronic Engineering
National Changhua University of Education
Email: tch@cc.ncue.edu.tw

2016/04/19

Basic Model of a MOSFET



Saturation region:

$$I_{ds} = K\frac{W}{L}\cdot\frac{(V_{gs}-V_t)^2}{2}$$

Linear region:

$$I_{ds} = K\frac{W}{L}[(V_{gs}-V_t)V_{ds}-\frac{V_{ds}^2}{2}]$$



For recent micron tech:

$$K_n \approx 2.5K_p \approx 100\mu A$$
$$V_t \approx 0.5V$$

A 10T+C 2-Stage Simple OP AMP



Half Operational Trans-conductance Amplifier



Swing ↑ ➔ W ↑

Av ↑ ➔ W_1 ↓ W_3 ↑ W_5 ↑

Voltage-Divider Biasing Circuit



$$I_{ds} = K_n\frac{W_n}{L_n}\cdot\frac{(V_{bias}-V_{tn})^2}{2} = K_p\frac{W_p}{L_p}\cdot\frac{(V_{DD}-V_{bias}-|V_{tp}|)^2}{2}$$

Assume: $L_n=L_p$, $K_n=2.5K_p$, $V_{tn}=|V_{tp}|=0.5V$, $V_{DD}=5V$,

$$W_p = \frac{2.5(V_{bias}-0.5)}{(4.5-V_{bias})}W_n$$

Differential Amplifier (Simple OTA)

$$r_L = \frac{L_1}{KW_1(v_{gs1}-V_t)}$$

$$A_v = -g_m r_L$$

$$g_{m3} = K_n\frac{W_3}{L_3}(v_{gs3}-V_t)$$



Typical design constraints
if all L are the same and $|V_{GS}|$ are similar:

$$I = K_n\frac{W_5}{L_5}\cdot\frac{(V_{bias}-V_{tn})^2}{2}$$

Simulation – Differential Amplifier (1)

```
Differential Amplifier
.lib 'mm0355v.l' tt
.global dd
Vdd     dd      0       5
.param  biasing =       2
Mp1     g       g       dd      dd      pch     L=0.35u W=2.5u
Mp2     o       g       dd      dd      pch     L=0.35u W=2.5u
Mn3     g       in+     m       m       nch     L=0.35u W=1u
Mn4     o       in-     m       m       nch     L=0.35u W=1u
Mn5     m       bias    0       0       nch     L=0.35u W=1u
Vbias   bias    0       biasing
VIN1    IN1     0       2.5
VIN2    IN2     0       2.5
Vin+    in+     IN1     AC      1
Vin-    in-     IN2     0
.AC     DEC     101     1       100G    SWEEP   biasing LIN     7       1       4
.plot
.options post
.end
```

1. Sweeping selected parameters
2. Optimization
3. Alter
4. C Srcript

Simulation – Differential Amplifier (2)



Level Shifter & Driver



The same as the homework 1.

Operating Points

Add **.OP** operating-point analysis

```
Differential Amplifier
.lib 'mm0355v.l' tt
.global dd
Vdd      dd       0        5
Mp1      g        g        dd       dd       pch      L=0.35u  W=2.5u
Mp2      o        g        dd       dd       pch      L=0.35u  W=2.5u
Mn3      g        in+      m        m        nch      L=0.35u  W=1u
Mn4      o        in-      m        m        nch      L=0.35u  W=1u
Mn5      m        bias     0        0        nch      L=0.35u  W=1u
Mp6      bias     bias     dd       dd       pch      L=0.35u  W=1u
Mn7      bias     bias     0        0        nch      L=0.35u  W=2.5u

VIN1     IN1      0        2.5
VIN2     IN2      0        2.5
Vin+     in+      IN1      AC       1
Vin-     in-      IN2      0
.AC      DEC      101      1        10G
.options post
.OP
.end
```

```
 ***** operating point status is all       simulation time is     0.
    node    =voltage      node    =voltage       node    =voltage

 +0:bias     =   1.3476  0:dd      =   5.0000  0:g       =   3.6991
 +0:in+      =   2.5000  0:in-     =   2.5000  0:in1     =   2.5000
 +0:in2      =   2.5000  0:m       =   1.4520  0:o       =   3.6991
```


Example

A 9T Non-Compensated OPAMP Design

```
Operational Amplifier
.lib 'mm0355v.l' tt
.global dd
Vdd     dd      0       5
Mp1     g       g       dd      dd      pch     L=0.35u W=2.5u
Mp2     o       g       dd      dd      pch     L=0.35u W=2.5u
Mn3     g       in+     m       m       nch     L=0.35u W=1u
Mn4     o       in-     m       m       nch     L=0.35u W=1u
Mn5     m       bias    0       0       nch     L=0.35u W=1u
Mp6     bias    bias    dd      dd      pch     L=0.35u W=1u
Mn7     bias    bias    0       0       nch     L=0.35u W=2.5u
Mp8     out     o       dd      dd      pch     L=0.35u W=4u
Mn9     out     bias    0       0       nch     L=0.35u W=1u

VIN1    IN1     0       2.5
VIN2    IN2     0       2.5
Vin+    in+     IN1     AC      1
Vin-    in-     IN2     0
.AC     DEC     101     1       100G
.options        post
.end
```


Example

A 9T OPAMP Design



Basic Negative Feedback System



Close loop transfer function: $T(s) = \frac{H(s)}{1+\beta H(s)}$

Barkhausen's criteria: $|\beta H| \le 1 @ \angle \beta H = -180° \Rightarrow \text{Stable}$

Typical Compensation on Bode Plot



Typical Frequency Compensation

1. Dominant-Pole Compensation
2. Pole-Zero Compensation
3. Miller Compensation (Pole-Splitting)

Miller Compensation

A 9T+C OPAMP



Miller Compensation

A 9T+C OPAMP

```
Operational Amplifier
.lib 'mm0355v.l' tt
.global dd
.param  Cap     =       0
Vdd     dd      0       5
Mp1     g       g       dd      dd      pch     L=0.35u W=2.5u
Mp2     o       g       dd      dd      pch     L=0.35u W=2.5u
Mn3     g       in+     m       m       nch     L=0.35u W=1u
Mn4     o       in-     m       m       nch     L=0.35u W=1u
Mn5     m       bias    0       0       nch     L=0.35u W=1u
Mp6     bias    bias    dd      dd      pch     L=0.35u W=1u
Mn7     bias    bias    0       0       nch     L=0.35u W=2.5u
Mp8     out     o       dd      dd      pch     L=0.35u W=4u
Mn9     out     bias    0       0       nch     L=0.35u W=1u
Cc      out     o       Cap

VIN1    IN1     0       2.5
VIN2    IN2     0       2.5
Vin+    in+     IN1     AC      1
Vin-    in-     IN2     0
.AC     DEC     101     10K     100G    SWEEP   Cap     DEC  2  1f  100p
.options        post
.end
```


Miller Compensation

A 9T+C OPAMP



Cross-Match Floor-Planning (1)

Basic Matching:

```
Mp1   g   g   dd   dd   pch   L=0.35u W=2.5u
Mp2   o   g   dd   dd   pch   L=0.35u W=2.5u
```



Cross-Match Floor-Planning (2)

Cross Matching:

```
Mp1   g   g   dd   dd   pch   L=0.35u  W=1.25u  M=2
Mp2   o   g   dd   dd   pch   L=0.35u  W=1.25u  M=2
```



Cross-Match Floor-Planning (3)

An Example for Coarse Floor-planning: